A COMPUTER PROGRAM FOR SIMULATION OF TAPERED DISPERSIONLESS LOSSY TRANSMISSION LINES

by

S. R. P. Teixeira and R. A. Rohrer

Memorandum No. ERL-M265

10 October 1969

E L E C T R O N I C S   R E S E A R C H   L A B O R A T O R Y

College of Engineering
University of California, Berkeley
94720

THE SUBROUTINE LINE

The FORTRAN IV subroutine LINE listed in the last section of this report compute the approximate response of a doubly loaded loss transmission line. The approximations employed are those which arise from representing the tapered line as a finite cascade of uniform lines and representing time only at equally spaced discrete points. Neither of these approximations need incur serious error since the user may trade granularity for computation time by increasing appropriate dimensions in the subroutine.

where:

x_1 is an integer constant or variable specifying the number of sections (it cannot be greater than 20 in value unless the subroutine is modified).

x_2 is a real constant or variable specifying the time step size for calculation. (As indicated above, this quantity must be smaller than the propagation time in any section (x_9), otherwise an error message, 'TIME STEP TOO LARGE', will be printed and execution stopped. For good precision x_2 should be as small as possible.)

x_3 is an integer constant or variable specifying the number of time steps (x_2) necessary to cover the total time interval of the solution (it cannot be greater than 30 unless the subroutine is modified; an x_3 greater than 30 causes the printing of an error message, 'TIME INTERVAL TOO LARGE' and stops execution).

x_4 is a real vector such that its ith element gives the value of the

Research sponsored by the Air Force Office of Scientific Research under Grant AF-AFOSR-1219-67 and the National Science Foundation under Grant GK-716.

voltage $e_0(t)$ for $t = (i - 1)x_2$.

x_5 is identical to x_4, but for $e_\lambda(t)$.

x_6 is a real constant or variable specifying the load resistance at the beginning of the line.

x_7 is a real constant or variable specifying the load resistance at the end of the line.

x_8 is a real vector such that its _i_th element is equal to $\sqrt{R_i G_i} \cdot \ell_i$, where R_i is the series resistance and G_i is the shunt conductance _per unit length_, respectively, and ℓ_i is the length of the _i_th line section.

x_9 is a real vector such that its _i_th element is equal to $\sqrt{L_i C_i} \cdot \ell_i$, where L_i is the series inductance and C_i is the shunt capacitance _per unit length_, respectively, and ℓ_i is the length of the _i_th line section.

x_{10} is a real vector such that its _i_th element is $\sqrt{L_i/C_i}$, the characteristic impedance of the _i_th section.

For consistency with the subroutine LINE as it stands, in the calling program, x_4 and x_5 must have dimensions 31 and x_8, x_9, and x_{10} must have dimensions 20.

The output of the subroutine LINE is the matrix x, where x is any real identifier, and the calling program must have the statement

COMMON/SLINE/x(2, 21, 31).

x(i, j, k) indicates a voltage if i = 1 or a current if i = 2 at the boundary point j (there are $x_1 + 1$ of these points) at time $t = (k - 1) \cdot x_2$.

If a number of sections x_1 or a number of time steps x_3 greater than the limits already stated are to be used, the _4_th, _5_th, and _6_th cards and the statement number 1 of the subroutine LINE must be modified by changing all 20's to the dimension of x_1, all 21's to the dimension of $x_1 + 1$, all 30's to the dimension of x_3, and all 31's to the dimension of $x_3 + 1$. The same rules must, of course, be applied to the COMMON/SLINE/ statement and to the DIMENSIONS of the appropriate vectors in the calling program.

SUBROUTINE LINE LISTING

```
      SUBROUTINE LINE(NSC,DEL,NI,VO,VL,RO,RL,A,V,Z)
C   SUBROUTINE FOR ANALYSIS OF TRANSMISSION LINES MADE UP OF A
C   FINITE NUMBER OF CONSTANT, DISPERSIONLESS LINE SECTIONS
      DIMENSION VO(31),VL(31),A(20),V(20),Z(20),R(2,20,31),IT(20)
      COMMON/SLINE/RF(2,21,31)
      IF(NSC.LE.20)GO TO 1
      PRINT 2
    2 FORMAT('   NUMBER OF SECTIONS TOO LARGE')
      STOP
    1 IF(NI.LE.30)GO TO 3
      PRINT 4
    4 FORMAT('      TIME INTERVAL TOO LARGE')
      STOP
    3 DO 5 I=1,NSC
      IF(DEL.GE.V(I))GO TO 6
    5 CONTINUE
      GO TO 7
    6 PRINT 8
    8 FORMAT('      TIME STEP TOO LARGE')
      STOP
    7 NI1=NI+1
      NSCM1=NSC-1
      DO 9 I=1,NSC
    9 IT(I)=V(I)/DEL
      DO 10 I=1,NI1
      II=I-IT(1)-1
      IF(II)11,12,13
   11 X=0.
      GO TO 15
   12 X=0.
      GO TO 14
   13 X=R(2,1,II)
   14 X=(R(2,1,II+1)-X)*(DEL*(IT(1)+1)-V(1))/DEL+X
   15 YR=(VO(I)+(RO/Z(1)-1.)*X)/(RO/Z(1)+1.)
      RF(1,1,I)=YR+X
      RF(2,1,I)=(YR-X)/Z(1)
      R(1,1,I)=YR
      IF(NSC.EQ.1)GO TO 16
      DO 17 J=1,NSCM1
      II=I-IT(J+1)-1
      IF(II)18,19,20
   18 X=0.
      GO TO 22
   19 X=0.
      GO TO 21
   20 X=R(2,J+1,II)
   21 X=(R(2,J+1,II+1)-X)*(DEL*(IT(J+1)+1)-V(J+1))/DEL+X
   22 II=I-IT(J)-1
      IF(II)23,24,25
```

```
   23 Y=0.
      GO TO 27
   24 Y=0.
      GO TO 26
   25 Y=R(1,J,II)
   26 Y=(R(1,J,II+1)-Y)*((IT(J)+1)*DEL-V(J))/DEL+Y
   27 E=EXP(-A(J))
      R(2,J,I)=(2.*Z(J)*X+(Z(J+1)-Z(J))*Y*E)*E/(Z(J+1)+Z(J))
      YR=(2.*Z(J+1)*Y*E+(Z(J)-Z(J+1))*X)/(Z(J)+Z(J+1))
      RF(1,J+1,I)=YR+X
      RF(2,J+1,I)=(YR-X)/Z(J+1)
   17 R(1,J+1,I)=YR
   16 II=I-IT(NSC)-1
      IF(II)28,29,30
   28 Y=0.
      GO TO 32
   29 Y=0.
      GO TO 31
   30 Y=R(1,NSC,II)
   31 Y=(R(1,NSC,II+1)-Y)*((IT(NSC)+1)*DEL-V(NSC))/DEL+Y
   32 E=EXP(-A(NSC))
      XR=(VL(I)+(RL/Z(NSC)-1.)*E*Y)*E/(1.+RL/Z(NSC))
      R(2,NSC,I)=XR
      Y=Y*E
      XR=XR/E
      RF(1,NSC+1,I)=XR+Y
   10 RF(2,NSC+1,I)=(Y-XR)/Z(NSC)
      RETURN
      END
```